APPROVED BY
J N Snyder

IDENTIFICATION

Binary Punch Routine, MU BPU1

E. M. Zographos - November 21, 1956

Midwestern Universities Research Association, Madison, Wisconsin

PURPOSE

To punch out a block of N words from magnetic core storage onto absolute binary cards. The loading address of the words is the same as the location of the words in storage.

RESTRICTIONS

This program will not properly punch itself out of the memory. However, the error made is an error in the check sum of a card. If the resultant check sum error stop is ignored on subsequent read-in of the MU BPU1 cards produced by MU BPU1, the program will be correctly loaded.

METHOD

Accuracy: Exact.

USAGE

Calling Sequence:

UA SAP

Loc.	Instruction
α	TSX BPU1, 4
$\alpha + 1$	PZE R, 0, N
$\alpha + 2$	Return

MURASS

Loc.	Instruction
α	BTSX A 4
$\alpha + 1$	C 0 R 0 N
$\alpha + 2$	Return

where:

R = initial location of the block of words in storage
N = number of words in the block

Space Required:

	UA SAP	MURASS
37 words program at:	BPU1	A
4 words temporary at:	COMMON	T

CODING INFORMATION

Timing: 905.4 ms. is the average time for the first card if the punch is not in motion upon entry. The punch will be kept in motion at full speed (100 cards/ min.) if the time between exit and re-entry does not exceed 24.6 ms.

If identifying punches in positions 9L1 - 9L12 are desired, card 37 of the MU BPU1 deck may be replaced by a card containing any octal number having digits in these positions. In the deck supplied to SHARE, card 37 is zero so that no identifying punches are generated. If this card were made

OCT 007400000000

then MURA binary cards (having 9L6 - 9L9 punches) would be the output.

```
        REM  BPU1 MURA BINARY PUNCH                                  MU   BPU1
BPU1    SXD  COMMON,1      STORE IR 1                                BPU1 0001
        SXD  COMMON+1,2    STORE IR2                                 BPU1 0002
        CLA  1,4           PROGRAM PARAMETER TO AC                   BPU1 0003
        PDX  0,1           SET IR1=N                                 BPU1 0004
        STO  COMMON+2      STORE R IN 9L ADDRESS                     BPU1 0005
        ARS  18            SHIFT N TO ADDRESS OF AC                  BPU1 0006
        ADD  COMMON+2      R+N                                       BPU1 0007
        STA  BPU1+11       PLANT ADDRESS FOR CKS LOOP                BPU1 0008
        WPU                SELECT PUNCH                              BPU1 0009
        LXA  BPU1+3,2      SET IR2=0                                 BPU1 0010
        PXD  0,0           CLEAR AC                                  BPU1 0011
        ACL  0,1           FORM CHECK SUM                            BPU1 0012
        TXI  BPU1+13,2,1   ADD 1 TO IR2                              BPU1 0013
        TXH  BPU1+15,2,21  TEST FOR FULL BINARY CARD                 BPU1 0014
        TIX  BPU1+11,1,1   REPEAT UNTIL DONE                         BPU1 0015
        SLW  COMMON+3      CHECK SUM TO 9R                           BPU1 0016
        PXD  0,2           V=NO. OF WORDS ON CARD                    BPU1 0017
        STD  COMMON+2      V TO 9L DECREMENT                         BPU1 0018
        ARS  18            SHIFT V TO ADDRESS OF AC                  BPU1 0019
        ADD  COMMON+2      R+V IN ADDRESS OF AC                      BPU1 0020
        STA  BPU1+28       PLANT ADDRESS FOR COPY LOOP               BPU1 0021
        CLA  BPU1+36       IDENTIFICATION                            BPU1 0022
        ORS  COMMON+2      COMPLETE 9L                               BPU1 0023
        CAL  COMMON+2      ADD 9L TO                                 BPU1 0024
        ACL  COMMON+3      PARTIAL CHECK SUM                         BPU1 0025
        SLW  COMMON+3      TOTAL CHECK SUM TO 9R                     BPU1 0026
        CPY  COMMON+2      COPY 9L                                   BPU1 0027
        CPY  COMMON+3      COPY 9R                                   BPU1 0028
        CPY  0,2           START OF COPY LOOP                        BPU1 0029
        TIX  BPU1+28,2,1   LOOP RE-ENTRY, OUT AFTER V PASSES         BPU1 0030
        CLA  BPU1+28       R+V IN ADDRESS                            BPU1 0031
        STA  COMMON+2      R+V TO 9L ADDRESS                         BPU1 0032
        TIX  BPU1+8,1,1    REPEAT UNTIL DONE                         BPU1 0033
        LXD  COMMON,1      RESTORE IR1                               BPU1 0034
        LXD  COMMON+1,2    RESTORE IR2                               BPU1 0035
        TRA  2,4           OUT                                       BPU1 0036
        HTR  0             CONSTANT                                  BPU1 0037
```